
History

• Lots of logical data models have been proposed in the history of 
DBMS
• Hierarchical (IMS), Network (CODASYL), Relational, etc

• What Goes Around Comes Around 
• Graph database uses data models that are “spiritual successors” of Network 

data model that is popular in 1970’s.
• CODASYL = Committee on Data Systems Languages

Supplier (sno, sname, scity) Supply (qty, price) Part (pno, pname, psize, pcolor)
supplies supplied_by



Edge-labelled Graph 

• We assign labels to edges that indicate the different types of relationships 
between nodes

• Nodes = {Steve Carell, The Office, B.J. Novak}
• Edges = {(Steve Carell, acts_in, The Office), (B.J. Novak, produces, The 

Office), (B.J. Novak, acts_in, The Office)}
• Basis of Resource Description Framework (RDF) aka. “Triplestore”



The Property Graph Model
• Extends Edge-labelled Graph with labels
• Both edges and nodes can be labelled with a set of property-value pairs 

attributes directly to each edge or node.
• The Office crew graph

• Node 𝑛" has node label Person with attributes: <name, Steve Carell>, 
<gender, male>
• Edge 𝑒" has edge label acts_in with attributes: <role, Michael G. Scott>, 

<ref, Wikipedia>



Property Graph v.s. Edge-labelled Graph
• Having node labels as part of the model can offer a more direct 

abstraction that is easier for users to query and understand
• Steve Carell and B.J. Novak can be labelled as Person

• Suitable for scenarios where various new types of meta-information 
may regularly need to be added to edges or nodes



Same Data, Different Model

• The same data represented in relational model

Redundant!



Schema Extensibility (and/or) Link Discovery
CREATE TABLE TVSHOW(title, year);
CREATE TABLE Production_Company(name, 

exec_prodcr)

graph: add an edge
SQL: redefine relational schema
ALTER TABLE TVSHOW(title, year, 

production_company references …);
-- foreign key constraint

Link Discovery: Data mine to 
determine new information

Property Graph v.s. Relational Model

• Graph Structure is more intuitive than a collection of tables (e.g., org 
chart)
• Ambiguity in graph representation using relational model (directed or 

undirected?)
• Avoid repetitive data storage from user perspective (e.g., primary key 

& foreign key)
• Enable same relation name with different attributes
• CREATE TABLE TVSHOW(title, year);
• CREATE TABLE TVSHOW(title, year, 
production_company); // Not possible!

• Nice query language for graph problems



Neo4j
• Neo4j is a graph database that uses property graph data model with a 

query language called Cypher
• In graph database domain, there is no standard query language (yet). 

Many vendor-dependent flavors
• SPARQL for RDF
• Cypher, Gremlin, etc. for property graph 
• Ex: Find co-stars of The Office

PREFIX : <http://ex.org/#>
SELECT ?x1 ?x2
WHERE {

?x1 :acts_in ?x3 . ?x1 :type :Person .
?x2 :acts_in ?x3 . ?x2 :type :Person .
?x3 :title "The Office" . ?x3 :type :TVSHOW .
FILTER(?x1 != ?x2)

}

MATCH (x1:Person) -[:acts_in]-> 
(:TVSHOW {title: "The Office"}) 
<-[:acts_in]- (x2:Person)

RETURN x1, x2

g.V().has("TVSHOW", "title", "The Office").
in('acts_in').hasLabel("Person").
values("name")



Graph Query Language (GQL)

• Ongoing standardization effort just like SQL for relational model



First Property Graph with Neo4j

• Demo: Create The Office crew graph in Neo4j

CREATE 
(n1:Person {name: "Steve Carell", gender: "male"}),
(n2:Person {name: "B.J. Novak", gender: "male"}),
(n3:TVShow {title: "The Office"}),
(n1)-[:acts_in {role: "Michael G. Scott", ref: "Wikipedia"}]->(n2),
(n2)-[:acts_in {role: "Ryan Howard", ref: "Wikipedia"}]->(n3),
(n2)-[:produces]->(n3);



Let’s Practice

• Let’s create the org. chart of the paper company Dunder Mifflin, 
Scranton Branch 1 in The Office.
• All edges have labels 𝑒$:𝑚𝑎𝑛𝑎𝑔𝑒𝑠 with 𝑖 being numbers from 1 to 𝑛, 

the number of edges
• Some useful commands & notes
• See the graph - MATCH (n) RETURN n LIMIT 50
• Delete the graph - MATCH (n) DETACH DELETE n
• To create list of values, use ”[]”

• For example, role: ["Sales", "Assistant Regional Manager"]

1. Season 4 Ep 16. Source: https://www.officetally.com/wp-content/uploads/2008/05/dunder_mifflin_org_chart.pdf



If some text is illegible, please reference
http://my.ilstu.edu/~llipper/com329/dunder_mifflin_org_chart.pdf



Graph Query Languages

• Two important usage patterns for graph query languages:
• Graph Pattern Matching
• Graph Navigation

• We’ll focus on Cypher in this tutorial. However, any significant graph 
query languages will have these two important patterns in their 
languages.



Graph Pattern Matching

• Graph Pattern Matching
• A match is a mapping from variables to constants such that when the 

mapping is applied to the given pattern, the result is, roughly speaking,  
contained within the original graph (i.e., subgraph).
• 𝑄": Find co-stars of The Office

graph pattern for 𝑄"

The Office Crew graph
Result set (i.e., matching) for 𝑄"



MATCH (x1:Person) -[:acts_in]-> (:TVSHOW {title: "The Office"}) <-[:acts_in]- (x2:Person)
RETURN x1, x2

Graph Pattern Matching in Cypher

• Cypher has no-repeated-edges, bags semantics
• 𝑄": Find co-stars of The Office

• Cypher manual: 
• https://neo4j.com/docs/cypher-manual/current/syntax/patterns/

Match pattern
We want to match variable 𝑥"to node with node label Person 

𝑥" has to connect to TVShow node through an incoming edge with edge label acts_in



Example

• Who’s inside Party Planning Committee (PPC)? (hint: PPC is a dept)

• How many people does Michael directly manage? (hint: use count())

MATCH (p:Person)<-[:manages]-(n:Person)
WHERE n.name = "Michael Scott"
RETURN count(p)

MATCH (p:Person)
WHERE "Party Planning Committee" in p.dept
return p.name



Let’s Practice

• Find all the employees that are directly managed by someone that 
reports to Michael

• Does Michael directly manage more employees than Jim Halpert?

MATCH (p {name: 'Michael Scott'})-[:manages]->()-[:manages]->(q)
RETURN q.name

MATCH (p:Person)<-[:manages]-(n:Person)
WHERE n.name = "Michael Scott"
WITH count(p) AS c1
MATCH (p:Person)<-[:manages]-(m:Person)
WHERE m.name = "Jim Halpert"
RETURN c1 > count(p)

Each MATCH ... WHERE can be thought as 
a SELECT ... FROM ... WHERE

MATCH (p:Person)<-[:manages]-(n:Person)
WHERE n.name = "Michael Scott"
MATCH (q:Person)<-[:manages]-(m:Person)
WHERE m.name = "Darryl Philbin"
RETURN p.name, q.name



Graph Navigation

• A mechanism provided by graph query languages to navigate the 
topology of the data.
• Two important query classes:
• Path Query
• Path Query + Graph Pattern Matching (i.e., navigational graph pattern)



Path Query

• Previously, we match a graph pattern; now, we match a path pattern.

• Path query has the general form 𝑃 = 𝑥→
0
𝑦 where 𝛼 specifies 

conditions on the paths we wish to retrieve and 𝑥 and 𝑦 are the 
endpoints of the path.

• 𝑄": Find co-stars of The Office

The Office Crew graph

𝑃 ≔ 𝑥
4567_$9 :4567_$9;

𝑦
Edge has direction!

Often represented 
using Regular 
Expressions



Path Query in Cypher

• Cypher has no-repeated-edge, bags semantics
• 𝑄": Find co-stars of The Office

• Nothing new but we return a path now!

MATCH path=(p:Person)-[:acts_in]->(:TVShow)<-[:acts_in]-(q:Person)
return path



Navigational Graph Pattern in Cypher

• We can combine path query with graph pattern matching by allowing 
edge labels in the graph pattern to be paths
• Q2: Find all the people that Michael Scott manages

• Resources: https://neo4j.com/docs/cypher-
manual/current/syntax/patterns/#cypher-pattern-relationship

MATCH path=(p:Person)-[:manages*1..]->(q:Person)
WHERE p.name = "Michael Scott"
return q.name



Example

• Get the Dunder Mifflin employees that are on the same level as 
“Michael Scott” (hint: use length()on path)
MATCH p1 = (n:Person)<-[:manages*]-(p:Person)
MATCH p2 = (m:Person)<-[:manages*]-(p:Person)
WHERE length(p1) = length(p2) AND m.name <> n.name AND n.name = "Michael Scott"
RETURN m



Same Data, Different Model

• Let’s query the same data in Relational Model

• Actual schema and data see “sql-ex-2.sql”



Same Data, Different Model

• Get the Dunder Mifflin employees that are on the same level as 
“Michael Scott”

with recursive samelevel(s1, s2, s3, s4) as (
(select a1.name, a1.mgrID, a2.name, a2.mgrID
from dunderMifflin a1, dunderMifflin a2
where a1.mgrID = a2.mgrID)
union
(select a1.name, a1.mgrID, a2.name, a2.mgrID
from dunderMifflin a1, dunderMifflin a2, samelevel l1
where a1.mgrID = l1.s2 and a2.mgrID = l1.s4)

) select l2.s3 from samelevel L2 where l2.s1 = 'Michael Scott' and l2.s1 <> l2.s3;

Base case: if two people are at the same level, 
their manager has to be the same.

Recursion: Same idea as 
base case but 
use the base relation and 
the result table we just 
computed in base case.

MATCH p1 = (n:Person)<-[:manages]-(p:Person)
MATCH p2 = (m:Person)<-[:manages]-(p:Person)
WHERE length(p1) = length(p2) AND m.name <> 
n.name AND n.name = "Michael Scott"
RETURN m



Let’s Practice

• Does Jim Halpert manage Phyllis Lapin?

• Find all people that are indirectly managed by Michael Scott (hint: use 
distinct)

MATCH path=(p:Person)-[:manages*1..]->(q:Person)
WHERE p.name = "Jim Halpert" and q.name = "Phyllis Lapin"
return count(path) > 0

MATCH path=(p1:Person {name: "Michael Scott"})-[:manages*1..]->()
-[:manages*1..]->(p2:Person)

return distinct p2

Why do we use distinct p2 
rather than distinct path?



Graph Algorithms in Cypher

• Cypher and many graph query languages allow user to directly embed 
graph algorithms inside the query
• Q3: Find the shortest path between David Wallace and Andy Bernard

• More graph algorithms: PageRank, Centrality, Connected Component 
algorithms, etc.

MATCH path = shortestPath(
(p:Person {name: "David Wallace"})-[:manages*1..]-(q:Person {name: "Andy Bernard"}))
RETURN path



Understand a Cypher Query

• Neo4j has EXPLAIN command; just like EXPLAIN in any RDBMS vendor
• Running example for this section
• Data modelled in Relational Model

• Query: SELECT * FROM r1 NATURAL JOIN r2 NATURAL JOIN r3;

CREATE TABLE R3(a char(1));
CREATE TABLE R2(a char(1), b integer);
CREATE TABLE R1(b integer);

INSERT INTO R3(a) VALUES (‘A’),(‘B’),(‘B’); 
INSERT INTO R2(a,b) VALUES 
(‘A’,1),(‘A’,1),(‘B’,1),(‘B’,2);
INSERT INTO R1(b) VALUES (2),(3);



Understand a Cypher Query (cont’)

• The plan for SQL query looks like



Understand a Cypher Query (cont’)

• Let’s model the same data in property graph model
• Guidelines to map relational model to graph model
• A row is a node
• A table name is a label name
• A join or foreign key is a relationship (i.e., edge)



Understand a Cypher Query (cont’)
Not So Correct Attempt

CREATE 
(n1:R3 {a: "A"}), 
(n2:R3 {a: "B"}), 
(n3:R3 {a: "B"}), 
(n4:R2 {a: "A", b: 1}), 
(n5:R2 {a: "A", b: 1}), 
(n6:R2 {a: "B", b: 1}), 
(n7:R2 {a: "B", b: 2}), 
(n8:R1 {b: 2}), 
(n9:R1 {b: 3}); 

MATCH (r1:R1),(r2:R2) 
CREATE (r1)-[:e]->(r2) 

MATCH (r2:R2),(r3:R3) 
CREATE (r2)-[:e]->(r3)

Violates a join or 
foreign key is a 
relationship: we 
have cartesian 
product



Understand a Cypher Query (cont’)
Better Attempt
CREATE 
(n1:R3 {a: "A"}), 
(n2:R3 {a: "B"}), 
(n3:R3 {a: "B"}), 
(n4:R2 {a: "A", b: 1}), 
(n5:R2 {a: "A", b: 1}), 
(n6:R2 {a: "B", b: 1}), 
(n7:R2 {a: "B", b: 2}), 
(n8:R1 {b: 2}), 
(n9:R1 {b: 3}); 

MATCH (r1:R1),(r2:R2)
WHERE r1.b = r2.b 
CREATE (r1)-[:e]->(r2) 

MATCH (r2:R2),(r3:R3)
WHERE r2.a = r3.a 
CREATE (r2)-[:e]->(r3)



Understand a Cypher Query (cont’)

• Resulting graph in Neo4j

• Observation
• Neo4j doesn't have notion of undirected edges during graph creation; user 

can ignore directions when they query the graph
• Unlike relational model, part of join computation is done during the graph 

model creation (e.g., when create relationships)



Understand a Cypher Query (cont’)

• Recall SQL query
• SELECT * FROM r1 NATURAL JOIN r2 NATURAL JOIN r3;

• In Cypher

• Let’s check the query plan
• EXPLAIN MATCH (r1:R1),(r2:R2),(r3:R3) WHERE r1.b = 
r2.b AND r2.a = r3.a RETURN r2.a, r1.b

MATCH (r1:R1),(r2:R2),(r3:R3) 
WHERE r1.b = r2.b AND r2.a = r3.a 
RETURN r2.a, r1.b



Understand a Cypher Query (cont’)
• Three-way join is computed using 

hash join in a pipeline fashion in 
Neo4j 4.2 optimizer

Suggests the above Cypher query is ”anti-pattern”



Understand a Cypher Query (cont’)

• Recall in HW4, we saw the semantics of semi-join in graph
• 𝐴 ⋉ 𝐵 means given a set of starting vertices in 𝐵, return the set of vertices in 
𝐴 that connect to those starting vertices via edges

• A multi-way join query = Graph Pattern Matching query
• Cypher query

MATCH (r3:R3)-[:e]-(r2:R2)-[:e]-(r1:R1) 
WHERE r2.a = r3.a and r2.b = r1.b 
RETURN r2.a, r1.b



Understand a Cypher Query (cont’)
• Expand(All)

• Given a start node, and depending on the pattern relationship, the 
Expand(All) operator will traverse incoming or outgoing relationships.

• = breadth-wise expansion of the search tree
• Join = Expand(All) + Filter

• The execution starts with 𝑟"and follows the edges from 𝑟" by one more level to 
find all 𝑟= that satisfies (r1)-[anon_27:e]-(r2)

• Then Filter is applied with predicate r2.b = r1.b. 
• Recall: Each MATCH ... WHERE can be thought as a SELECT ... FROM ... 

WHERE

MATCH (r2:R2)-[:e]-(r3:R3) 
WHERE r2.a = r3.a 
with r2 
MATCH (r2)-[:e]-(r1:R1) 
WHERE r2.b = r1.b 
RETURN r2.a, r1.b



Relational vs. Graph: A Case Study

• A term project done by a student in the same class in Spring 2019
• The goal is to enable run queries against a large collection of data 

(~2.2M rows) gathered from student’s workplace
• Example query: Where are all my photos?
• Similar scenarios:

• Query against files stored in S3 bucket
• Query against log files collected over time from different services on AWS

• Method: model data in relational model (Postgres 11) and graph 
model (Neo4j 3.5) and compare performance
• Independent of the DBMS, the raw data comprised a labeled directed 

graph



Relational vs. Graph: A Case Study (cont’)

• Query 1: Exact Filename Match
• Return full paths of files that match a given file name

Cypher query is much shorter* than 
SQL counterpart à easier development 
and code maintenance

*average < 1/3 LoC. Newer, similar 
studies show even greater reduction 
(e.g., occasionally < 1/10 LoC)



Relational vs. Graph: A Case Study (cont’)
• Postgres outperforms Neo4j for certain queries

• Cypher query still needs many hand tuning* 
to reach acceptable performance à indicates 
lack of maturity for Neo4j optimizer

*writing query in a different way, and/or   
overriding Neo4j’s Cypher Query Optimizer’s 
choice of query plan.



Conclusion

• Introduced Edge-label Graph, Property Graph
• Discussed their difference with each other and with Relational Model

• Introduced graph query languages 
• SPARQL for RDF (i.e., Edge-label Graph), Gremlin and Cypher for Property Graph
• Introduced three important usage patterns in graph query languages

• Graph Pattern Matching
• Path Query
• Navigational Graph Pattern Matching

• Demonstrated and practiced those usage patterns in Cypher with Neo4j
• Introduced Cypher query profiling in Neo4j

• A real world case study on relational model vs. graph model
• Graph query is easier to write and maintain
• Neo4j has a long way to go to build a sophisticated optimizer like Postgres



Moving Forward

• Gremlin
• https://kelvinlawrence.net/book/Gremlin-Graph-Guide.html
• https://tinkerpop.apache.org/docs/current/tutorials/getting-started/

• Contrast among Cypher, SQL, and Datalog on the same data
• https://github.com/xxks-kkk/Code-for-blog/tree/master/2020/sql-datalog-

cypher
• Code for this tutorial
• https://github.com/xxks-kkk/Code-for-blog/tree/master/2020/intro-to-

graphdb-with-neo4j
• Slides available
• https://zhu45.org/introduction-to-graph-database-with-neo4j.pdf

https://kelvinlawrence.net/book/Gremlin-Graph-Guide.html
https://tinkerpop.apache.org/docs/current/tutorials/getting-started/
https://github.com/xxks-kkk/Code-for-blog/tree/master/2020/sql-datalog-cypher
https://github.com/xxks-kkk/Code-for-blog/tree/master/2020/intro-to-graphdb-with-neo4j
https://zhu45.org/introduction-to-graph-database-with-neo4j.pdf

